The procedures of *OreModules* can be used to treat underdetermined linear systems of PDEs with variable coefficients. We present one example that appears in mathematical physics, namely the study of Lie-Poisson structures. See C. M. Bender, G. V. Dunne, L. R. Mead, *Underdetermined systems of partial differential equations*, Journal of Mathematical Physics, vol. 41, no. 9 (2000), pp. 6388-6398 and W. M. Seiler, *Involution analysis of the partial differential equations characterising Hamiltonian vector fields*, Journal of Mathematical Physics, vol. 44 (2003), pp. 1173-1182.

```
> with(Ore_algebra):
> with(OreModules):
```

For the computations that follow, we define the Weyl algebra $Alg = A_3$: In this Ore algebra we have for $1 \le i,\, j \le 3$, $\mathrm{D}_i\, x_j = x_j\, \mathrm{D}_i + \delta_{i,\,j}$, where $\delta_{i,\,j}$ is the Kronecker symbol.

```
> Alg := DefineOreAlgebra(diff=[D[1],x[1]], diff=[D[2],x[2]],
> diff=[D[3],x[3]], polynom=[x[1],x[2],x[3]]):
```

The following example appears in (Bender, Dunne, Mead, 2000), where the E_2 algebra is studied. We want to parametrize the solutions of the linear system of partial differential equations which is defined by the following matrix:

```
> R := evalm([[x[1]*D[3], x[2]*D[3], 0],
> [-x[1]*D[2]+x[2]*D[1], -1, x[2]*D[3]],
> [-1, -x[2]*D[1]+x[1]*D[2], x[1]*D[3]]]);
```

$$R := \begin{bmatrix} x_1\,\mathrm{D}_3 & x_2\,\mathrm{D}_3 & 0 \\ -x_1\,\mathrm{D}_2 + x_2\,\mathrm{D}_1 & -1 & x_2\,\mathrm{D}_3 \\ -1 & -x_2\,\mathrm{D}_1 + x_1\,\mathrm{D}_2 & x_1\,\mathrm{D}_3 \end{bmatrix}$$

In order to check parametrizability, we compute the formal adjoint of R first:

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -x_1\,\mathrm{D}_3 & -x_2\,\mathrm{D}_1 + x_1\,\mathrm{D}_2 & -1 \\ -x_2\,\mathrm{D}_3 & -1 & -x_1\,\mathrm{D}_2 + x_2\,\mathrm{D}_1 \\ 0 & -x_2\,\mathrm{D}_3 & -x_1\,\mathrm{D}_3 \end{bmatrix}$$

We compute the first extension module with values in *Alg* of the left *Alg*-module which is associated with *R_adj*:

```
> Ext1 := Exti(R_adj, Alg, 1);
```

$$Ext1 := \left[\begin{bmatrix} \mathrm{D}_3 & 0 & 0 \\ -x_2\,\mathrm{D}_1 + x_1\,\mathrm{D}_2 & 0 & 0 \\ 0 & x_2\,\mathrm{D}_3 & 0 \\ 0 & x_1\,\mathrm{D}_3 & 0 \\ 0 & -x_2\,\mathrm{D}_1 + x_1\,\mathrm{D}_2 & 0 \\ 0 & 0 & 1 \end{bmatrix},\ \begin{bmatrix} x_1 & x_2 & 0 \\ \mathrm{D}_1 & \mathrm{D}_2 & \mathrm{D}_3 \\ -1 & -x_2\,\mathrm{D}_1 + x_1\,\mathrm{D}_2 & x_1\,\mathrm{D}_3 \end{bmatrix},\ \begin{bmatrix} -x_2\,\mathrm{D}_3 \\ x_1\,\mathrm{D}_3 \\ -x_1\,\mathrm{D}_2 + x_2\,\mathrm{D}_1 \end{bmatrix} \right]$$

Since *Ext1*[1] is not an identity matrix, we see that the above system of linear PDEs is *not* parametrizable. The rows of *Ext1*[2] give a generating set of torsion elements of the left *Alg*-module M associated to this system. The non-zero elements in the ith column of *Ext1*[1] generate the annihilator of the ith row r_i of *Ext1*[2] in M, i.e. the set of elements P of *Alg* that satisfy $P\, r_i = 0$ in M. The generating set of torsion elements of M can be displayed in a more familiar way by using *TorsionElements*:

```
> T := TorsionElements(R, [F(x[1],x[2],x[3]), G(x[1],x[2],x[3]),
> H(x[1],x[2],x[3])], Alg);
```

$$T := \left[\left[\begin{array}{c} \frac{\partial}{\partial x_3} \%2 = 0 \\ -x_2 (\frac{\partial}{\partial x_1} \%2) + x_1 (\frac{\partial}{\partial x_2} \%2) = 0 \\ x_2 (\frac{\partial}{\partial x_3} \%1) = 0 \\ x_1 (\frac{\partial}{\partial x_3} \%1) = 0 \\ -x_2 (\frac{\partial}{\partial x_1} \%1) + x_1 (\frac{\partial}{\partial x_2} \%1) = 0 \end{array} \right], \right.$$

$$\left. \left[\begin{array}{c} \%2 = x_1 \mathrm{F}(x_1, x_2, x_3) + x_2 \mathrm{G}(x_1, x_2, x_3) \\ \%1 = (\frac{\partial}{\partial x_1} \mathrm{F}(x_1, x_2, x_3)) + (\frac{\partial}{\partial x_2} \mathrm{G}(x_1, x_2, x_3)) + (\frac{\partial}{\partial x_3} \mathrm{H}(x_1, x_2, x_3)) \end{array} \right] \right]$$

$$\%1 := \theta_2(x_1, x_2, x_3)$$
$$\%2 := \theta_1(x_1, x_2, x_3)$$

We exhibit the same parametrization as in (Bender, Dunne, Mead, 2000) (up to the mistake in (Bender, Dunne, Mead, 2000) concerning the existence of the torsion elements which is underlined in (Seiler, 2003)). In fact, *Ext1*[3] is a parametrization of the torsion-free part $M \ / \ \mathrm{t}(M)$ of M:

```
> ApplyMatrix(Ext1[3], [H(x[1],x[2],x[3])], Alg);
```

$$\left[\begin{array}{c} -x_2 (\frac{\partial}{\partial x_3} \%1) \\ x_1 (\frac{\partial}{\partial x_3} \%1) \\ x_2 (\frac{\partial}{\partial x_1} \%1) - x_1 (\frac{\partial}{\partial x_2} \%1) \end{array} \right]$$
$$\%1 := \mathrm{H}(x_1, x_2, x_3)$$

Since M is not a torsion module, i.e. $\mathrm{t}(M) \neq M$, but R is a square matrix, we notice that R does not have full row rank. The relations that the rows of R satisfy are computed in a free resolution of M:

```
> Free := FreeResolution(R, Alg);
```

$$\mathit{Free} := \mathrm{table}([1 = \left[\begin{array}{ccc} x_1 \mathrm{D}_3 & x_2 \mathrm{D}_3 & 0 \\ -x_1 \mathrm{D}_2 + x_2 \mathrm{D}_1 & -1 & x_2 \mathrm{D}_3 \\ -1 & -x_2 \mathrm{D}_1 + x_1 \mathrm{D}_2 & x_1 \mathrm{D}_3 \end{array} \right],$$
$$2 = \left[\begin{array}{ccc} -x_2 \mathrm{D}_1 + x_1 \mathrm{D}_2 & x_1 \mathrm{D}_3 & -x_2 \mathrm{D}_3 \end{array} \right],$$
$$3 = \mathrm{INJ}(1)$$
$$])$$

We see that the left *Alg*-module L associated to *Free*[2] is torsion-free:

```
> ext := Exti(Involution(Free[2], Alg), Alg, 1);
```

$$\mathit{ext} := \left[\left[\begin{array}{c} 1 \end{array} \right], \left[\begin{array}{ccc} -x_2 \mathrm{D}_1 + x_1 \mathrm{D}_2 & x_1 \mathrm{D}_3 & -x_2 \mathrm{D}_3 \end{array} \right], \left[\begin{array}{ccc} \mathrm{D}_3 & x_1 \mathrm{D}_3 & 0 \\ -\mathrm{D}_2 & -x_1 \mathrm{D}_2 + x_2 \mathrm{D}_1 & x_2 \\ -\mathrm{D}_1 & -1 & x_1 \end{array} \right] \right]$$

The system associated to *Free*[2] is parametrizable because L is torsion-free. Of course, $R = \mathit{Free}[1]$ is a parametrization of *Free*[2], but we have found another parametrization in *ext*[3]. We compute the rank of L:

```
> OreRank(Free[2], Alg);
```

$$2$$

We see that neither R nor *ext*[3] is a minimal parametrization of *Free*[2]. Let us compute some minimal parametrizations of *Free*[2]:

```
> MinimalParametrizations(Free[2], Alg);
```

$$\left[\begin{bmatrix} D_3 & x_1 D_3 \\ -D_2 & -x_1 D_2 + x_2 D_1 \\ -D_1 & -1 \end{bmatrix}, \begin{bmatrix} D_3 & 0 \\ -D_2 & x_2 \\ -D_1 & x_1 \end{bmatrix}, \begin{bmatrix} x_1 D_3 & 0 \\ -x_1 D_2 + x_2 D_1 & x_2 \\ -1 & x_1 \end{bmatrix}\right]$$

Let us finish by checking whether the system given by R is parametrizable or not if we turn the variables x_1, x_2, x_3 into invertible elements of *Alg*. More precisely, we consider the system to be given over the Ore algebra B_3, which differs from the Weyl algebra A_3 only in the fact that the domain of coefficients of D_1, D_2, D_2 is not the polynomial ring in x_1, x_2, x_3 but the field of rational functions in x_1, x_2, x_3. However, using *OreModules*, we use $Alg = A_3$ as above, but instead of *Exti* we apply *ExtiRat*, which takes into account that the coefficients are rational functions:

```
> st := time():  Ext1 := ExtiRat(R_adj, Alg, 1); time() - st;
```

$$Ext1 := \left[\begin{bmatrix} D_3 & 0 \\ -x_1 D_2 + x_2 D_1 & 0 \\ 0 & D_3 \\ 0 & -x_1 D_2 + x_2 D_1 \end{bmatrix}, \begin{bmatrix} x_1 & x_2 & 0 \\ 0 & -x_2 + x_1 x_2 D_1 - D_2 x_1^2 & -x_1^2 D_3 \end{bmatrix}, \begin{bmatrix} x_2 D_3 \\ -x_1 D_3 \\ -x_2 D_1 + x_1 D_2 \end{bmatrix}\right]$$

$$0.231$$

Again, we conclude that the system is not parametrizable because non-trivial torsion elements exist. A generating set of torsion elements is given in *Ext1*[2] and can also be obtained in this case by using *TorsionElementsRat*:

```
> TorsionElementsRat(R, [F(x[1],x[2],x[3]), G(x[1],x[2],x[3]),
> H(x[1],x[2],x[3])], Alg);
```

$$\left[\begin{bmatrix} \frac{\partial}{\partial x_3} \%3 = 0 \\ x_2 \left(\frac{\partial}{\partial x_1} \%3\right) - x_1 \left(\frac{\partial}{\partial x_2} \%3\right) = 0 \\ \frac{\partial}{\partial x_3} \%1 = 0 \\ x_2 \left(\frac{\partial}{\partial x_1} \%1\right) - x_1 \left(\frac{\partial}{\partial x_2} \%1\right) = 0 \end{bmatrix}, \begin{bmatrix} \%3 = x_1 F(x_1, x_2, x_3) + x_2 \%2 \\ \%1 = -x_2 \%2 + x_1 x_2 \left(\frac{\partial}{\partial x_1} \%2\right) - x_1^2 \left(\frac{\partial}{\partial x_2} \%2\right) - x_1^2 \left(\frac{\partial}{\partial x_3} H(x_1, x_2, x_3)\right) \end{bmatrix}\right]$$

$\%1 := \theta_2(x_1, x_2, x_3)$

$\%2 := G(x_1, x_2, x_3)$

$\%3 := \theta_1(x_1, x_2, x_3)$